*==========Class_D_Resistor==========

It looks like a class D resistor can be made by sensing output current and adjusting duty cycle to match the V to I relationship desired. In this case a 5 Ohm resistor is built. The output load resistor has been turned into a small sense resistor RS. A transconductance stage is being used to offset the duty cycle from 50%.

This effective 5 Ohm resistor resembles a real resistor in terms of voltage, current and the ability to absorb power. But the power that is getting absorbed is being transferred at high efficiency to the power supplies.




```
*.OPTIONS  GMIN=1f       METHOD=trap   ABSTOL=1u     TEMP=27   srcsteps = 1  gminsteps = 1
*.OPTIONS  RELTOL=.001   ABSTOL=1n     VNTOL=1u      ITL4=500  ITL1=400
*==========Create_Signal==================
VT         VT     0      DC     0      PWL(   0      0    1      1)
Vfreq      Vfreq  0      DC     200k
VD         VD     0      DC     .5
RD         VD     VD2    1000k
CCOMP      VD2    0      50pF   IC=.5
B_OTA      VD2    0      I      = 1m*(v(out)-v(is) )
VPI        VPI    0      DC     3.141592653589793
B_TRI      TRI    0      V =    acos( cos(6.283185*V(VFreq)*V(VT)) )/v(VPI)
BVG        VG     0      V =    5*u( v(TRI) -v(VD2)) -2.5
VCC        VCC    0      DC     2.5
VEE        VEE    0      DC     -2.5
RP         VCC    VP     1u
RPF        VP     VPF    10K
CPF        VPF    0      .001u
RN         VN     VEE    1u
RNF        VN     VNF    10K
CNF        VNF    0      .001u
MN1        VOUT   VG     VN     VEE    NMOSC   W=90000u    L=1u
MP1        VOUT   VG     VP     VCC    PMOSC   W=90000u    L=1u
```

```
L1          VOUT    OUT     100u
C1          OUT     0       3u
RS          OUT     IS      1m
Rout        IS      VCM     .1
Vrout       Vrout   0       DC      .1
VCM         VCM     0       DC      0       SIN(    0    2.5    50       )
*TRAN       TSTEP   TSTOP   TSTART TMAX     ?UIC?
.tran        .1u     20m      1m      .1u    UIC
*==========The_CMOS_Model_Files=====================================
.model      NMOSC                   NMOS(Level=   1   Cbs=2f Cbd=2f)
.model      PMOSC                   PMOS(Level=   1   Cbs=2f Cbd=2f)
.control
run
linearize
set         pensize = 2
plot        vcm out
plot        vd2
```

==========

Because the LCR tuned circuit at the output has two poles around 9Khz, the "resistor" has a limited bandwidth. But 50Hz is no problem.





==========

And one is going to want to do a V to I curve.

```
let      iout = -(out-vcm)/.1
plot     out vs iout
```



==========

The current in the CMOS transistors can be observed with a little low pass filtering.

```
plot      vcc vpf mean(vp)
```



==========

So now power in VCC, VEE, and VCM can be monitored.

```
let       vccpwrac=vcc*(vcc-vpf)/1u
let       veepwrac=vee*(vee-vnf)/1u
let       vcmpwrac=-vcm*i(vcm)
let       supplypwrac= vccpwrac+ veepwrac
plot      vccpwrac veepwrac vcmpwrac
```



==========

The powers in VCC an VEE together will match and the power of VCM in magnitude, but with opposite polarity.

```
plot      vcmpwrac supplypwrac
```



==========

And the power out of VCM is what one would expect from a 5 Ohm resistor.

plot norm(vcm) norm(vcmpwrac)



So now AC power can be converged to DC power in a much different way.

```
============Full_Netlist_For_Copy_Paste=========================
ClassD_Resistor
*
*
*              __________________________________
*       _____ |                                  |
*      _|_   | ||\                               |
*     /   \  | |_|+\           RB                |
*    /B_TRI\ |   |  \______                      |                     CCOMP
*    \     / |   |  /      |    _/\  /\  /\___|______________________
*     \___/  |___|-/ BVG   |   |    \/  \/   CCOMP              |    _|_
*      _|_       |/        |   |                                |    ___
*      ///                 |  _|_                _____|\  BOTA   |     |
*      ____________________|  /VB \             |    |-\ _  _   |    _|_
*      |                      \___/             |    |  \/ \/ \__|    ///
*      |                       _|_              |    |  /\_/\_/
*      |                       ///              |  _|+/
*      |  VCC _/\  /\  /\_ VP                   |  | |/
*      |     _|_  \/  \/   |                    |  |
*      |    /VCC\ RP      <-            ___     |  |     ___
*      |    \___/      ||____ MP1      |OUT|_|  |____|IS |
*      |     _|_    ____||             |___|          |___|
*      |     ///   |   ||_                  |   RS=1m
*      |           |     |       _  _  _    |              |
*      |___________|   _|_|VOUT|__| ()  () |_|__/\  /\  /\_|____
*  ____        VG  |___||  |____|     L1    |     \/  \/      _|_
* _|_ |              ||_                    |                /   \ VCM
*  /// _|_           ||-> MN1  Freq = 9 K   |       C1      // \  \
*    /VEE\    RN       |       Z=5.8Ohms    |___||__ 3u     \    \//
*    \___/ _/\  /\  /\_|                        ||  _|_       \___/
*      |___|  \/  \/  VN                            ///        _|_
*                                                               ///*
*.OPTIONS  GMIN=1f       METHOD=trap   ABSTOL=1u     TEMP=27    srcsteps = 1  gminsteps = 1
*.OPTIONS  RELTOL=.001   ABSTOL=1n     VNTOL=1u      ITL4=500   ITL1=400
*=========Create_Signal==================
VT        VT     0      DC      0      PWL(   0      0    1      1)
Vfreq     Vfreq  0      DC      200k
VD        VD     0      DC      .5
RD        VD     VD2    1000k
CCOMP     VD2    0      50pF    IC=.5
B_OTA     VD2    0      I       = 1m*(v(out)-v(is) )
VPI       VPI    0      DC      3.141592653589793
B_TRI     TRI    0      V =     acos( cos(6.283185*V(VFreq)*V(VT)) )/v(VPI)
BVG       VG     0      V =     5*u( v(TRI) -v(VD2)) -2.5
VCC       VCC    0      DC      2.5
VEE       VEE    0      DC      -2.5
RP        VCC    VP     1u
RPF       VP     VPF    10K
CPF       VPF    0      .001u
RN        VN     VEE    1u
RNF       VN     VNF    10K
CNF       VNF    0      .001u
MN1       VOUT   VG     VN      VEE    NMOSC   W=90000u    L=1u
MP1       VOUT   VG     VP      VCC    PMOSC   W=90000u    L=1u
L1        VOUT   OUT    100u
C1        OUT    0      3u
RS        OUT    IS     1m
Rout      IS     VCM    .1
Vrout     Vrout  0      DC      .1
VCM       VCM    0      DC      0      SIN(   0    2.5   50      )

*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
.tran      .1u    20m     1m     .1u   UIC

*=========The_CMOS_Model_Files===================================
.model    NMOSC               NMOS(Level=  1  Cbs=2f Cbd=2f)
.model    PMOSC               PMOS(Level=  1  Cbs=2f Cbd=2f)

.control
run
linearize
```

```
set       pensize = 2
plot      vcm out
plot      vd2

let       iout = -(out-vcm)/.1
plot      out vs iout
plot      vcc vpf mean(vp)

let      vccpwrac=vcc*(vcc-vpf)/1u
let      veepwrac=vee*(vee-vnf)/1u
let      vcmpwrac=-vcm*i(vcm)
let      supplypwrac= vccpwrac+ veepwrac

plot     vccpwrac veepwrac vcmpwrac
plot     vcmpwrac supplypwrac
plot     norm(vcm) norm(vcmpwrac)

.endc
.end

9.16.10_2.06PM
dsauersanjose@aol.com
Don Sauer
```